Lagrangian Interpolation—LAIN

The program LAIN (**La**grangian **In**terpolation) is a subroutine subprogram that interpolates the given equally spaced data at the equidistant dividing points using the Lagrange's four-point method. The application of this subroutine is limited to the case where the data are given as equally spaced points on the axes, otherwise the axes must be transformed once so that the given data become equally spaced data. Example 2 is an example of such a case.

LAIN (**La**grangian **In**terpolation)

【Purpose】

To interpolate the given equally spaced data with equally spaced division points using Lagrange's four-point method.

【Usage】

(1) How to connect

```
CALL LAIN (DX, NN, A, ND1, AA, ND2, NDIV)
```

Argument	Type	Parameter in calling program	Return Parameter
DX	R	Interval of original data	Interval of interpolated data
NN	I	Total number of original data	Total number of interpolated data
A	R 1-D array (ND1)	Original equally spaced data	Unchanged
ND1	I	Dimension size of A in calling program	Unchanged
AA	R 1-D array (ND2)	No need to input here	Interpolated equally spaced data
ND2	I	Dimension size of AA in calling program	Unchanged
NDIV	I	Number of subdivision (NDIV .LE. 20)	Unchanged

(2) Necessary subroutines and function subprograms

None

(3) Remarks

i) *ND2*, the dimension size of AA, must be *ND2* ≥ (*NN* -1)×*NDIV* + 1.

ii) Interpolation may change the maximum value of the data, so correct it if necessary.

【Calculation Method】

According to the Lagrange's interpolation formula, the function $g(x)$ given the values $g(x_k)$ ($k = 1, 2, 3, \cdots, n+1$) at $n+1$ points $x_n = 1, 2, 3, \cdots, n+1$, respectively, can be approximated by the following nth order polynomial $f(x)$.

$$f(x)=\sum_{k=1}^{n+1} g(x_k)\,\frac{(x-x_1)(x-x_2)\cdots(x-x_{k-1})(x-x_{k+1})\cdots(x-x_{n+1})}{(x_k-x_1)(x_k-x_2)\cdots(x_k-x_{k-1})(x_k-x_{k+1})\cdots(x_k-x_{n+1})} \qquad \text{(a)}$$

The right-hand side of Eq. (a) is a polynomial of degree n with respect to x. If $x = x_k$, it is clear that $f(x_k) = g(x_k)$. The function $f(x)$ coincides with the function $g(x)$ at $x_n = 1, 2, 3, \cdots, n+1$.

Now, if we set $n = 3$ in Eq. (a), that is, if we use the four-point method to give the value of the function $g(x)$ at four points, Eq. (a) becomes the following.

$$\begin{aligned} f(x) = {} & g(x_1)\frac{(x-x_2)(x-x_3)(x-x_4)}{(x_1-x_2)(x_1-x_3)(x_1-x_4)} \\ & + g(x_2)\frac{(x-x_1)(x-x_3)(x-x_4)}{(x_2-x_1)(x_2-x_3)(x_2-x_4)} \\ & + g(x_3)\frac{(x-x_1)(x-x_2)(x-x_4)}{(x_3-x_1)(x_3-x_2)(x_3-x_4)} \\ & + g(x_4)\frac{(x-x_1)(x-x_2)(x-x_3)}{(x_4-x_1)(x_4-x_2)(x_4-x_3)} \end{aligned} \qquad \text{(b)}$$

Assuming that x_1, x_2, x_3, and x_4 have equal spacing Δx, we can set the following.

$$x_1 = -\Delta x,\ x_2 = 0,\ x_3 = \Delta x,\ x_4 = 2\Delta x$$

Substituting the above relationship into Eq. (b), we get

$$\begin{aligned} f(x) = {} & g(x_1)\frac{(x-0)(x-\Delta x)(x-2\Delta x)}{(-\Delta x)(-2\Delta x)(-3\Delta x)} \\ & + g(x_2)\frac{(x+\Delta x)(x-\Delta x)(x-2\Delta x)}{(\Delta x)(-\Delta x)(-2\Delta x)} \\ & + g(x_3)\frac{(x+\Delta x)(x-0)(x-2\Delta x)}{(2\Delta x)(\Delta x)(-\Delta x)} \\ & + g(x_4)\frac{(x+\Delta x)(x-0)(x-\Delta x)}{(3\Delta x)(2\Delta x)(\Delta x)} \end{aligned} \qquad \text{(c)}$$

Let $\xi = x/\Delta x$ as shown in the following figure, and write $g(x_1)$, $g(x_2)$, $g(x_3)$, and $g(x_4)$ as g_1, g_2, g_3, and g_4. Then, Eq. (c) becomes

$$f(\xi) = c_1(\xi)g_1 + c_2(\xi)g_2 + c_3(\xi)g_3 + c_4(\xi)g_4 \qquad \text{(d)}$$

where

$$\left.\begin{aligned} c_1 &= \frac{(\xi-0)(\xi-1)(\xi-2)}{(-1)(-2)(-3)} \\ c_2 &= \frac{(\xi+1)(\xi-1)(\xi-2)}{(+1)(-1)(-2)} \\ c_3 &= \frac{(\xi+1)(\xi-0)(\xi-2)}{(+2)(+1)(-1)} \\ c_4 &= \frac{(\xi+1)(\xi-0)(\xi-1)}{(+3)(+2)(+1)} \end{aligned}\right\} \qquad \text{(e)}$$



If we assume that the interval Δx of the given data is to be divided into equally spaced *ndiv* segments, then the first half of the program, we compute Eq. (e) for ξ of (3*ndiv* +1) between ξ = -1 and ξ = 2. Then we compute $f(\xi)$ in Eq. (d) at *ndiv* points between ξ = 0 and 1 by sequentially equating g_1, g_2, g_3, and g_4 to the four subsequent values of the given data. The program that performs the latter part of the calculation is rather complicated. The reason is that it also handles the irregular interpolations in the first and last intervals of the given data together.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * *
C     SUBROUTINE FOR LAGRANGEAN INTERPOLATION
C  * * * * * * * * * * * * * * * * * * * * * * *
C
C     PURPOSE
C       TO INTERPOLATE THE GIVEN, EQUI-SPACED DATA BY LAGRANGE'S
C       FOUR-POINTS METHOD
C
C     USAGE
C       CALL LAIN(DX,NN,A,ND1,AA,ND2,NDIV)
C
C     DESCRIPTION OF ARGUMENTS
C       DX      - INTERVAL OF ORIGINAL/INTERPOLATED DATA
C                 AT CALL/RETURN
C       NN      - TOTAL NUMBER OF ORIGINAL/INTERPOLATED DATA
C                 AT CALL/RETURN  NN(AT CALL).GE.4
C       A(ND1)  - ORIGINAL EQUI-SPACED DATA
C       ND1     - DIMENSION OF A IN CALLING PROGRAM
C       AA(ND2) - INTERPOLATED EQUI-SPACED DATA
```

```
C       ND2     - DIMENSION OF AA IN CALLING PROGRAM
C                 ND2.GE.NN(AT CALL)*NDIV
C       NDIV    - NUMBER OF SUBDIVISION  NDIV.LE.20
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C       NONE
C
      SUBROUTINE LAIN(DX,NN,A,ND1,AA,ND2,NDIV)
C
      DIMENSION A(ND1),AA(ND2),XD(4),D(4),C(80,4),AD(4)
      DATA      XD/-1.,0.,1.,2./,D/-6.,2.,-2.,6./
C
      DIV=REAL(NDIV)
      I4=NDIV*3+1
      DO 130 I=1,I4
      X=REAL(I-NDIV-1)/DIV
      DO 120 J=1,4
      P=1.
      DO 110 K=1,4
      IF(K.EQ.J) GO TO 110
      P=P*(X-XD(K))
  110 CONTINUE
      C(I,J)=P/D(J)
  120 CONTINUE
  130 CONTINUE
      N=0
      IS=1
      IE=NDIV*2
      DO 170 M=2,NN-2
      DO 140 J=1,4
      AD(J)=A(M+J-2)
  140 CONTINUE
      DO 160 I=IS,IE
      AA1=0.
      DO 150 J=1,4
      AA1=AA1+AD(J)*C(I,J)
  150 CONTINUE
      N=N+1
      AA(N)=AA1
  160 CONTINUE
      IF(M.EQ.2) IS=NDIV+1
      IF(M.EQ.NN-3) IE=I4
  170 CONTINUE
      NN=N
      DX=DX/DIV
      RETURN
      END
```

【Example1】

Interpolate the data shown in the table below by dividing it into four parts to obtain a time history with a time interval of 0.5/4 = 0.125 seconds.

Table Sample data

No	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
$T=m\Delta t$	0.0	0.5	1.0	1.5	2.0	2.5	3.0	3.5	4.0	4.5	5.0	5.5	6.0	6.5	7.0	7.5
x_m	5	32	38	-33	-19	-10	1	-8	-20	10	-1	4	11	-1	-7	-2

```
C
      DIMENSION A(16),AA(64)
      DATA  DT/0.5/,NN/16/,A/5.0,32.0,38.0,-33.0,-19.0,-10.0,1.0,-8.0,-20.0,
     *              10.0,-1.0,4.0,11.0,-1.0,-7.0,-2.0/,NDIV/4/
C
      CALL LAIN(DT,NN,A,16,AA,64,NDIV)
      STOP
      END
```

Output: The thin line and circle dots are the given data, and the thick blue line is the interpolated time history.



【Example2】

Assume that the table data used in Example1 is given by $T_k = 16/k$ second and interpolate each time interval into three parts.

The time point T_k at which the data is given is not equally spaced, but if we take the reciprocal of this, we get equally spaced data with an interval of 1/16. Therefore, first we can perform such a transformation, then interpolate data by LAIN, and finally take the reciprocal of T_r again to restore the time.

```
C
      DIMENSION A(16),AA(48),T(48)
      DATA  NN/16/,A/5.0,32.0,38.0,-33.0,-19.0,-10.0,1.0,-8.0,-20.0,
     *              10.0,-1.0,4.0,11.0,-1.0,-7.0,-2.0/,NDIV/3/
C
      DT=1.0/REAL(NN)
      DT1=DT
      CALL LAIN(DT,NN,A,16,AA,48,NDIV)
      DO 110 K=1,NN
      TR=REAL(K-1)*DT+DT1
      T(K)=1.0/TR
  110 CONTINUE
      STOP
      END
```

Output: The result is represented by the data in the array *AA* for each time stored in the array T. The plot is as follows. The thin line and circle dots are the given data, and the thick blue line is the interpolated time history. The horizontal axis uses a logarithmic scale.